
Рекомендация МСЭ-Т G.107 

 Е-модель – вычислительная модель, используемая при планировании передачи 

 

 

Резюме 

В данной Рекомендации приведен алгоритм для так называемой Е-модели в качестве общей модели 
МСЭ-Т для оценки передачи речи. Эта вычислительная модель может быть полезна проектировщикам 
передачи, чтобы убедиться в том, что пользователи будут удовлетворены сквозным качеством 
передачи. Первоначально выходом этой модели служила скалярная оценка качества передачи. 
Главной особенностью модели является использование коэффициентов снижения качества, которые 
отражают воздействия современных устройств обработки сигнала. 

Для лучшего учета шума помещения на стороне передачи, а также искажений квантования в 
исправленном издании 2000 года был предложен улучшенный вариант Е-модели. Совместно с 
изданием 2002 года, снижение качества из-за случайной потери пакетов было учтено для различных 
кодеков параметрическим способом. Начиная с издания 2003 года, применялось улучшенное 
моделирование качества для случая низкого местного эффекта говорящего. Текущее издание 
обеспечивает более точное предсказание качества для кодеков при (кратковременной) зависимой 
потере пакетов. 

 

 

Источник 

Рекомендация МСЭ-Т G.107 утверждена 1 марта 2005 года 12-й Исследовательской комиссией МСЭ-Т 
(2005–2008 гг.) в соответствии с процедурой, изложенной в Рекомендации МСЭ-Т A.8. 

ПРЕДИСЛОВИЕ 

Международный союз электросвязи (МСЭ) является специализированным учреждением Организации 
Объединенных Наций в области электросвязи. Сектор стандартизации электросвязи МСЭ (МСЭ-Т) – 
постоянный орган МСЭ. МСЭ-Т отвечает за изучение технических, эксплуатационных и тарифных вопросов и 
за выпуск Рекомендаций по ним с целью стандартизации электросвязи на всемирной основе. 

Всемирная ассамблея по стандартизации электросвязи (ВАСЭ), которая проводится каждые четыре года, 
определяет темы для изучения Исследовательскими комиссиями МСЭ-Т, которые, в свою очередь, 
вырабатывают Рекомендации по этим темам.  

Утверждение Рекомендаций МСЭ-Т осуществляется в соответствии с процедурой, изложенной в Резолюции 1 
ВАСЭ. 

В некоторых областях информационных технологий, которые входят в компетенцию МСЭ-Т, необходимые 
стандарты разрабатываются на основе сотрудничества с ИСО и МЭК. 

 

 

 
ПРИМЕЧАНИЕ 

В настоящей Рекомендации термин "администрация" используется для краткости и обозначает как 
администрацию электросвязи, так и признанную эксплуатационную организацию. 

Соблюдение положений данной Рекомендации носит добровольный характер. Однако в Рекомендации могут 
содержаться определенные обязательные положения (например, для обеспечения возможности взаимодействия 
или применимости), и соответствие данной Рекомендации достигается в случае выполнения всех этих 
обязательных положений. Для выражения необходимости выполнения требований используется синтаксис 
долженствования и соответствующие слова (такие, как "должен" и т. п.), а также их отрицательные 
эквиваленты. Использование этих слов не предполагает, что соблюдение положений данной Рекомендации 
является обязательным для какой-либо из сторон. 

 

 

 
ПРАВА ИНТЕЛЛЕКТУАЛЬНОЙ СОБСТВЕННОСТИ 

МСЭ обращает внимание на то, что практическое применение или реализация этой Рекомендации может 
включать использование заявленного права интеллектуальной собственности. МСЭ не занимает какую бы то ни 
было позицию относительно подтверждения, обоснованности или применимости заявленных прав 
интеллектуальной собственности, независимо от того, отстаиваются ли они членами МСЭ или другими 
сторонами вне процесса подготовки Рекомендации. 

На момент утверждения настоящей Рекомендации МСЭ не получил извещение об интеллектуальной 
собственности, защищенной патентами, которые могут потребоваться для реализации этой Рекомендации. 
Однако те, кто будет применять Рекомендацию, должны иметь в виду, что это может не отражать самую 
последнюю информацию, и поэтому им настоятельно рекомендуется обращаться к патентной базе данных БСЭ. 

 

 

 

  ITU   2005 

Все права сохранены. Никакая часть данной публикации не может быть воспроизведена с помощью каких-либо 
средств без письменного разрешения МСЭ. 

Рекомендация МСЭ-Т G.107 

 Е-модель – вычислительная модель, используемая при планировании передачи 

1 Общие положения 

1.1 Область применения 

В данной Рекомендации описана вычислительная модель, известная как Е-модель, которая признана 
полезным инструментом при планировании передачи сигналов для оценки комбинированных 
эффектов изменения различных параметров передачи, влияющих на качество переговоров по 
аппаратам с микротелефонными трубками в спектре телефонии 3,1 кГц1. Например, эта 
вычислительная модель может быть использована проектировщиками передачи, чтобы убедиться в 
том, что пользователи будут удовлетворены сквозным качеством передачи, избегая в то же время 
усложнений проектирования сетей. Следует подчеркнуть, что первоначально выходом этой модели 
был "коэффициент рейтинга" ("Rating Factor") R, но этот коэффициент может быть преобразован, 
чтобы дать оценку мнения пользователя. Такие оценки сделаны только для целей планирования 
передачи, а не для предсказания мнения (для которого не существует согласованной модели, 
рекомендованной МСЭ-Т) конкретного потребителя. 

Теперь в это исправленное издание в качестве нового параметра включена потеря пакетов и 
улучшенное моделирование местного эффекта говорящего. 

Для очень большого числа возможных комбинаций входных параметров Е-модель еще полностью не 
проверена практикой эксплуатации или лабораторными исследованиями. Для многих комбинаций, 
имеющих значение для проектировщиков передачи, Е-модель может быть использована с некоторой 
степенью доверия, однако для других комбинаций параметров прогнозы Е-модели проверяются и 
сейчас находятся на стадии изучения. Соответственно, при некоторых обстоятельствах следует 
проявлять осторожность при использовании Е-модели. Например, Е-модель может дать неточные 
результаты для комбинаций определенных видов снижения качества. Дополнительная информация по 
этому вопросу приведена в Приложении A. 

1.2 Ссылки 

Указанные ниже Рекомендации МСЭ-Т и другие источники содержат положения, которые путем 
ссылки на них в данном тексте составляют положения настоящей Рекомендации. На момент 
публикации указанные издания были действующими. Все Рекомендации и другие источники могут 
подвергаться пересмотру; поэтому всем пользователям данной Рекомендации предлагается изучить 
возможность применения последнего издания Рекомендаций и других источников, перечисленных 
ниже. Список действующих в настоящее время Рекомендаций МСЭ-Т регулярно публикуется. Ссылка 
на документ в данной Рекомендации не придает ему, как отдельному документу, статус 
Рекомендации. 

[1] ITU-T Recommendation G.100 (2001), Definitions used in Recommendations on general 
characteristics of international telephone connections and circuits. 

[2] ITU-T Recommendation G.108 (1999), Application of the E-model: A planning guide. 

[3] ITU-T Recommendation G.109 (1999), Definition of categories of speech transmission quality. 

[4] Рекомендация МСЭ-Т G.113 (2001), Ухудшение передачи из-за обработки речи. 

[5] ITU-T Recommendation G.113 Appendix I (2002), Provisional planning values for the equipment 
impairment factor Ie and packet-loss robustness factor Bpl. 

____________________ 
1 В данном контексте качество переговоров относится к характеристикам передачи, например, 

продолжительности передачи, эффектам эха говорящего и т. п. Однако описанная в этой Рекомендации 
E-модель не предназначена для моделирования снижения качества передачи в ситуации одновременного 
разговора. 

[6] ITU-T Recommendation P.833 (2001), Methodology for derivation of equipment impairment factors 
from subjective listening-only tests. 

[7] ITU-T Recommendation P.834 (2002), Methodology for the derivation of equipment impairment 
factors from instrumental models. 

[8] ITU-T Recommendation P.862 (2001), Perceptual evaluation of speech quality (PESQ): An 
objective method for end-to-end speech quality assessment of narrow-band telephone networks and 
speech codecs. 

2 Е-модель – вычислительная модель, используемая при планировании передачи 

2.1 Введение 

Сложность современных сетей требует, чтобы при планировании передачи многие параметры не 
только учитывались индивидуально, но также принимались в расчет действия их комбинаций. Это 
может быть сделано "экспертом, обладающим предугадыванием", но желателен более 
систематический подход, как, например, использующий вычислительную модель. Выходом описанной 
здесь модели является скалярная величина рейтинга качества R, которая изменяется пропорционально 
общему качеству разговора. В Рекомендации МСЭ-Т G.113 [4] дано (основанное на упрощении 
модели) руководство по особым видам снижения качества, включая эффекты от комбинаций. Однако 
выходом могут также быть номинальные оценки реакций пользователей, например в форме 
процентов, которыми оценивают моделированное соединение: "хорошо или лучше" или "плохо или 
хуже", как это описано в Приложении B. Кроме того, подробное руководство для соответствующего 
приложения Е-модели − как это описано в данной Рекомендации − приведено в Рекомендации МСЭ-
Т G.108 [2]. Дополнительно определение категорий качества передачи можно найти в Рекомендации 
МСЭ-Т G.109 [3]. 

2.2 Исходная программа 

В Приложении C содержится исходная программа на языке BASIC для Е-модели, описанной в данной 
Рекомендации. Цель этой программы – гарантировать пользователям Е-модели непротиворечивое 
применение аналитических выражений. 

3 Структура и основа алгоритмов Е-модели 

Е-модели основаны на методе коэффициентов снижения качества оборудования, за которым следуют 
предварительные модели рейтинга передачи. Метод был разработан специальной группой ETSI, 
которая называлась "Качество передачи речи ото рта к уху" ("Voice Transmission Quality from Mouth to 
Ear"). 

Эталонное соединение, показанное на рисунке 1, разделено на передающую и приемную сторону. 
Модель оценивает качество разговора ото рта к уху, как это воспринимается пользователем на 
приемной стороне в качестве слушающего и говорящего. 

Ожидаемый коэффициент А 

Рисунок 1/G.107 – Эталонное соединение Е-модели 

Параметры передачи, использованные на входе вычислительной модели, показаны на рисунке 1. 
Величины шума помещения и коэффициентов D в алгоритме вычисления обрабатывают отдельно для 
передающей и приемной сторон, и они могут иметь разные значения. Параметры SLR, RLR и шум 
цепи Nc сравнивают с определенной точкой 0 дБо. Все прочие входные параметры рассматривают 
либо как значения для общего соединения, например OLR (в некоторых случаях сумма SLR и RLR), 
число устройств qdu, коэффициенты снижения качества оборудования Ie и коэффициент выигрыша A, 
либо как значения, относящиеся только к приемной стороне, например STMR, LSTR, WEPL (для 
вычисления эха слушающего) и TELR. 

Существуют три различных параметра, связанные со временем передачи. Абсолютная задержка Ta 
представляет общую задержку в одном направлении между передающей и приемной сторонами, 
которую используют для оценки снижения качества из-за слишком большой задержки. Значение 
параметра задержки в одном направлении T представляет задержку между приемной стороной (в 
состоянии разговора) и точкой соединения, в которой объединение сигналов создает источник эха. 
Задержка в оба конца Tr представляет только задержку в 4-проводной замкнутой цепи, в которой 
"дважды отраженный" сигнал вызывает снижение качества из-за эха слушающего. 

3.1 Вычисление коэффициента рейтинга передачи R 

Согласно методу вычисления коэффициента снижения качества оборудования, фундаментальный 
принцип Е-модели основан на концепции, которая приведена в описании модели OPINE 
[см. Литература, Supplement 3 to P-series]: 

Психологические коэффициенты аддитивны в психологическом масштабе. 

На первом этапе любого вычисления с Е-моделью результатом является коэффициент рейтинга 
передачи R, который объединяет все параметры передачи, имеющие отношение к этому соединению. 
Этот коэффициент рейтинга R состоит из следующих составляющих: 

  AIe-effIdIsRoR +−−−= . (3-1) 

В принципе Ro представляет основное соотношение сигнал/шум, включая такие источники шума, как 
шум цепи и шум помещения. Коэффициент Is представляет комбинацию из всех снижений качества, 
которые действуют на речевой сигнал более или менее одновременно. Коэффициент Id представляет 
снижение качества, вызываемое задержкой, а коэффициент снижения эффективности оборудования 
Ie-eff представляет снижение качества, вызываемое кодеками с низкой битовой скоростью. Этот 
коэффициент также включает снижение качества из-за потери пакетов с произвольным 
распределением. Коэффициент выигрыша A позволяет компенсировать коэффициенты снижения 
качества в тех случаях, когда существуют другие преимущества доступа к пользователю. Член Ro и 
величины Is и Id подразделяют на дальнейшие специфические значения снижения качества. 
В следующих разделах представлены формулы, использованные в Е-модели. 

3.2 Основное отношение сигнал/шум Ro 

Основное отношение сигнал/шум Ro определено как: 

  ( )NoSLRRo +−= 5,115 . (3-2) 

Член No [в дБм0п] представляет сложение по мощности от различных источников шума: 

  











+++= 10101010 10101010log10

NfoNorNosNc

No . (3-3) 

Член Nc [в дБм0п] представляет сумму всех источников мощности шума цепи относительно точки 0 
дБо. 

Член Nos [в дБм0п] представляет эквивалентный шум цепи в точке 0 дБо, вызванный шумом 
помещения Ps на передающей стороне: 

  ( )214004,0100 −−−+−−−= DsOLRPsDsSLRPsNos , (3-4) 

где OLR = SLR + RLR. Таким же образом шум помещения Pr на приемной стороне преобразуют в 
эквивалентный шум цепи Nor [в дБм0п] в точке 0 дБо. 

  2)35(008,0121 −++−= PrePreRLRNor . (3-5) 

Член Pre [в дБм0п] представляет "эффективный шум помещения", вызванный увеличением на Pr за 
счет цепи местного эффекта слушающего пользователя: 

  











++= 10

)–10(

101log10
LSTR

PrPre . (3-6) 

Nfo [в дБм0п] представляет "нижний уровень шума" на приемной стороне: 

  RLRNforNfo += . (3-7) 
с обычно принимаемым значением Nfor, равным −64 дБмп. 

3.3 Коэффициент одновременного снижения качества Is 

Коэффициент Is представляет сумму всех снижений качества, которые могут произойти более или 
менее одновременно при передаче речи. В дальнейшем коэффициент Is подразделяют на три 
специфических коэффициента снижения качества: 

  IqIstIolrIs ++= . (3-8) 

Коэффициент Iolr представляет вызванное слишком низкими значениями OLR снижение качества, 
которое определяют как: 

  





































+=

8
–

8
120

8
1

8 XolrXolrIolr , (3-9) 

где: 

  )64(2,0 RLRNoOLRXolr −++= . (3-10) 

Коэффициент Ist представляет снижение качества, вызванное не оптимальным местным эффектом: 

29
33

3–113
4,19

1128–
6

13–112
13
1

1335
1

358
1

8

+

















+−



















 ++

















+= STMRoSTMRoSTMRoIst , (3-11) 

где: 

  











+=

−−−
10410 10e10log10–

TELRTSTMR

STMRo . (3-12) 

Коэффициент снижения качества Iq представляет снижение качества, вызванное искажениями 
квантования: 

  [ ]ZYIq 10101log15 ++= , (3-13) 

где: 

  
94,8

46
15

100 GRY o −+−=  (3-14) 

  
4030

46 GZ −=  (3-15) 

и: 

  20602,0258,007,1 QQG ++=  (3-16) 

  )qdu(log1537 −=Q . (3-17) 

В этой формуле qdu означает число устройств qdu для всего соединения между передающей стороной 
и приемной стороной. 
ПРИМЕЧАНИЕ. – Если коэффициент снижения качества Ie используют для части оборудования, то 
величину qdu нельзя использовать для той же самой части оборудования. 

3.4 Коэффициент снижения качества из-за задержки Id 

Коэффициент снижения качества Id, также представляющий все снижения качества из-за задержки 
речевых сигналов, далее подразделяют на три коэффициента Idte, Idle и Idd: 

  IddIdleIdteId ++= . (3-18) 

Коэффициент Idte дает оценку снижения качества из-за эха говорящего: 

  ( )TReRoeReRoeIdte –
2

e–11–100
4

)–(
2
–












++= , (3-19) 

где: 

  )(5,1 RLRNoRoe −−=  (3-20) 

  )14(5,280 −+= TERVRe  (3-21) 

23,0–e6

150
1

10
1

log40– T

T

T

TELRTERV +
+

+
= . (3-22) 

Для значений T < 1 мс эхо говорящего следует рассматривать как местный эффект, то есть Idte = 0. 
Следовательно, вычислительный алгоритм объединяет влияние STMR на эхо говорящего. Принимая 
во внимание, что низкие значения STMR могут создавать для эха говорящего некоторые 
маскирующие эффекты, а для очень высоких значений STMR эхо говорящего может стать более 
заметным, члены TERV и Idte выбирают следующим образом: 

Для STMR < 9 дБ: 

в уравнении 3-21 TERV заменяют на TERVs, где: 

  
2

IstTERVTERVs += . (3-23) 

Для 9 дБ ≤ STMR ≤ 20 дБ: 

применяют вышеприведенные уравнения от 3-19 до 3-22. 

Для STMR > 20 дБ: 

В уравнении 3-18 Idte заменяют на Idtes, где: 

  22 IstIdteIdtes += . (3-24) 

Коэффициент Idle представляет снижение качества из-за эха слушающего. Для этого используют 
уравнения: 

  169
4

)–(
2
– 2

++= RleRoRleRoIdle , (3-25) 

где: 

  25,0–)1)(7(5,10 ++= TrWEPLRle  (3-26) 

Коэффициент Idd представляет снижение качества, вызванное длительной абсолютной задержкой Ta, 
которая имеет место даже при очень хорошем подавлении эха. 

Для Ta ≤ 100 мс: 

  0=Idd . 

Для Ta > 100 мс: 

  ( )
















+













++= 2

3
13–125

6
1

6
6
1

6 XXIdd , (3-27) 

где: 

  
2log

100
log 








=

Ta

X . (3-28) 

3.5 Коэффициент снижения качества оборудования Ie 

Значения коэффициента Ie снижения качества оборудования из-за использования элементов 
низкоскоростных кодеков не связаны с другими входными параметрами. Эти значения зависят от 
результатов тестирования по субъективной усредненной оценке (mean opinion score), а также от опыта 
работы в сети. Реальные рекомендованные значения Ie приведены в Дополнении I/G.113 [5]. 

Ранее, при произвольной потере пакетов2, особые значения коэффициента снижения качества кодека 
интерпретировали, пользуясь табулированными значениями Ie, зависящими от потери пакетов. Теперь 
в качестве специального значения кодека определен коэффициент устойчивости к потере пакетов Bpl. 
Зависящий от потери пакетов коэффициент снижения эффективности оборудования Ie-eff находят, 
используя характерный для кодека коэффициент снижения качества оборудования при нулевой потере 
пакетов Ie и коэффициент устойчивости к потере пакетов Bpl. Для некоторых кодеков оба 
коэффициента приведены в Дополнении I/G.113. При вероятности потери пакетов Ppl 
коэффициент Ie-eff вычисляют по формуле: 

  
Bpl

BurstR
Ppl

PplIeIeIe-eff
+

⋅−+= )95( . (3-29) 

BurstR – это так называемый коэффициент всплеска (Burst Ratio), который определяют как: 

ельностипоследоват йпринимаемо в выбросовх наблюдаемыдлина  Средняя  =BurstR  

Если потеря пакетов произвольная (то есть независимая), BurstR = 1;  
если потеря пакетов со всплеском (то есть зависимая), BurstR > 1. 

Например, для потери пакетов с распределением, соответствующим модели Маркова (Markov) с двумя 
состояниями и с вероятностями p переходов от состояния "обнаружен" к состоянию "потерян" и с 
вероятностями q – от состояния "потерян" к состоянию "обнаружен", коэффициент всплеска может 
быть вычислен как: 

  
q

Ppl
p

Ppl
qp

BurstR 100/1100/1 −==
+

= . (3-30) 

Как видно из формулы 3-29, коэффициент снижения эффективности оборудования в случае Ppl = 0 
(отсутствия потери пакетов) равен значению Ie, определенному в Дополнении I/G.113.  

Информация о значениях параметров, для которых действителен этот алгоритм, приведена в 
Приложении A/G.107. 

3.6 Коэффициент выигрыша A 

Благодаря особому значению коэффициента выигрыша A в конечном результате не существует 
никаких его связей с другими параметрами передачи. Некоторые предварительные значения 
приведены в таблице 1. 

Таблица 1/G.107 – Предварительные примеры для коэффициента выигрыша A 

Пример системы связи Максимальное значение A 

Обычная (проводная) 0 
Мобильная с помощью сотовой сети в здании 5 
Мобильная в географической области или подвижная на транспортном 
средстве 

10 

Доступ к труднодоступному месту, например с помощью спутников со 
многими интервалами соединений 

20 

____________________ 
2 Вероятность потери пакета полагают не зависящей от состояния приема (принят или потерян) предыдущего 

пакета. 

Средняя длина выбросов, ожидаемых в сети с "произвольной" потерей пакетов 

Следует отметить, что значения в таблице 1, взятые из Рекомендации МСЭ-Т G.113 [4], являются 
только предварительными. Использование коэффициента A и его подобранных значений в 
специальных приложениях оставлено на усмотрение проектировщика. Однако значения в таблице 1 
следует рассматривать как абсолютный верхний предел для A. 

3.7 Значения по умолчанию 

Для всех входных параметров, использованных в Е-модели, значения по умолчанию приведены в 
таблице 2. Настоятельно рекомендуется использовать значения по умолчанию для всех параметров, 
которые не изменяются во время проектных расчетов. Если все параметры установлены на значения 
по умолчанию, результаты вычислений покажут очень высокое качество с коэффициентом рейтинга 
R = 93,2. 

Таблица 2/G.107 – Значения по умолчанию и разрешенные диапазоны параметров 

Параметр Аббрев. Единицы
Значение  

по 
умолчанию 

Разрешен- 
ный 

диапазон 
Замечание 

Рейтинг громкости передачи SLR дБ  +8 0 ... +18 (Прим. 1)
Рейтинг громкости приема RLR дБ  +2 −5 ... +14 (Прим. 1)

Рейтинг маскировки местного эффекта STMR дБ  15 10 ... 20 (Прим. 2)
Рейтинг местного эффекта слушающего LSTR дБ  18 13 ... 23 (Прим. 2)
Значение D телефона на передающей стороне Ds –  3 –3 ... +3 (Прим. 2)
Значение D телефона на приемной стороне Dr –  3 −3 ... +3 (Прим. 2)

Рейтинг громкости эха говорящего TELR дБ  65 5 ... 65  
Взвешенное затухание канала эха WEPL дБ  110 5 ... 110  
Средняя задержка канала эха в одном 
направлении  

T мс  0 0 ... 500  

Задержка в двух направлениях в 4-проводной 
замкнутой цепи 

Tr мс  0 0 ... 1000  

Абсолютная задержка в соединениях, 
свободных от эха 

Ta мс  0 0 ... 500  

Число устройств с искажением квантования qdu –  1 1 ... 14  
Коэффициент снижения качества оборудования Ie –  0 0 ... 40  
Коэффициент устойчивости к потере пакетов Bpl –  1 1 ... 40 (Прим. 3)
Вероятность случайной потери пакетов Ppl %  0 0 ... 20 (Прим. 3)
Коэффициент всплеска BurstR –  1 1 … 2 (Прим. 3)
Шум цепи относительно точки 0 дБо Nc дБм0п −70 −80 ... −40  

Пороговый шум на стороне приема Nfor дБмп −64 – (Прим. 3)

Шум помещения на стороне передачи Ps дБ(A)  35 35 ... 85  
Шум помещения на стороне приема Pr дБ(A)  35 35 ... 85  
Коэффициент выигрыша A –  0 0 ... 20  
ПРИМЕЧАНИЕ 1. – Общие величины между микрофоном или приемником и точкой 0 дБо. 
ПРИМЕЧАНИЕ 2. – Постоянное отношение: LSTR = STMR + D. 
ПРИМЕЧАНИЕ 3. – В настоящее время изучается. 

Исправленная версия этой Рекомендации 2000 года предоставила улучшенный вариант алгоритма Е-
модели (см. Приложение A). 

Благодаря этой исправленной версии 2000 года результирующий рейтинг R при всех значениях 
параметров, принятых по умолчанию, несколько изменился (от R = 94,2 до R = 93,2). Однако для целей 
практического проектирования это небольшое различие следует считать незначительным. 

Приложение A 
 

Условия использования Е-модели 

ПРИМЕЧАНИЕ. − Оценка и улучшение алгоритма Е-модели оставлены для дальнейшего изучения. 
Новые результаты будут включены, как только они станут доступными. 

A.1 Примеры условий, при которых следует проявлять осторожность в использовании Е-
модели 

− Общий уровень коэффициентов снижения качества оборудования 
 Некоторые экспериментальные исследования показывают, что общая тенденция для 

коэффициентов снижения качества оборудования излишне пессимистична, так что на самом 
деле может существовать некоторый невидимый запас. 

− Общее аддитивное свойство модели 
 Полагают, что в Е-модели различные виды снижения качества аддитивны в масштабе 

коэффициента рейтинга передачи R. Это свойство пока не проверено до такой степени, чтобы 
сделать удовлетворительный вывод. Особенно мало исследовалось взаимодействие 
низкоскоростных кодеков с другими источниками снижения качества, например с шумом 
помещения. Кроме того, остается неопределенным порядок сложения эффектов при 
каскадном соединении низкоскоростных кодеков. 

− Действие местного эффекта говорящего  
 Некоторые эксперименты показывают, что Е-модель игнорирует некоторые маскирующие 

эффекты, которые случаются в местном эффекте говорящего, а именно связанные с шумом 
цепи, шумом помещения на приемной стороне и незначительной задержкой эха говорящего 
(< 10 мс). 

− Коэффициент выигрыша A 
 До сих пор остается неясным, при каких условиях могут быть использованы заданные 

величины для коэффициента выигрыша. Предполагают, что эти величины могут зависеть, 
например, от группы пользователя и будут изменяться в течение длительного периода. 

− Методология получения коэффициентов снижения качества для нового оборудования 
 Новая методология получения коэффициентов снижения качества оборудования, основанная 

на субъективной оценке качества при прослушивании, была одобрена как Рекомендация 
МСЭ-Т P.833 [6]. Новая методология для вычисления коэффициентов снижения качества 
оборудования по инструментальной модели, подобной P.862 [8], одобрена как Рекомендация 
МСЭ-Т P.834 [7]. 

− Предсказания для разных типов шума помещения и различных форм амплитудно-частотных 
характеристик в канале связи, в трактах местного эффекта и эха 

 Е-модель учитывает эффект шума помещения только с помощью взвешенного уровня A. 
Действительное мнение о качестве речевой связи может зависеть даже от типа и 
возмущающего воздействия окружающего шума. Амплитудно-частотные характеристики 
канала связи, трактов местного эффекта и эха не воспринимаются в явной форме Е-моделью, а 
только косвенно с помощью рейтингов громкости. Однако они могут влиять на 
воспринимаемое качество передачи. 

A.2 Условия, при которых характеристики Е-модели улучшают обновлением ее прежней 
версии 

− Эффект шума помещения на передающей стороне 
 В улучшенном алгоритме Е-модели (исправленная версия 2000 года) эффект Ломбарда 

(Lombard) (явление, при котором говорящий адаптируется к его/ее произношению и уровню 
речи относительно окружающего шума) более не игнорируется. Это имело место в версии 
1998 года и приводило к излишне пессимистическим предсказаниям Е-модели для высоких 
уровней шума помещения Pr. 

− Предсказания искажений квантования 
 Для Е-модели версии 1998 года результаты субъективных тестов, выраженные в значениях 

MNRU, очень часто оказывались более пессимистичными, чем предсказания по Е-модели 
версии 2000 года. Графики на рисунке A.1 построены по версии 1998 года и исправленной 
версии 2000 года, причем все параметры имеют значения по умолчанию. 

G.107_FA.1

Рейтинг R E-модели согласно Рекомендации МСЭ-Т 
G.107 (12/1998) 
Рейтинг R E-модели согласно Рекомендации МСЭ-Т 
G.107 (05/2000)  

Рисунок A.1/G.107 − Зависимость между числом устройств qdu  
и рейтингом R Е-модели  

Что касается несколько улучшенного алгоритма Е-модели, приведенного в данной 
Рекомендации, то зависимость между параметром qdu и рейтингом R Е-модели была изменена 
таким образом, чтобы лучше подогнать алгоритм под результаты субъективных тестов. 

− Предсказания характеристик кодека при стохастической потере пакетов 
 Первоначально в условиях потерь пакетов снижение качества из-за кодеков оценивали с 

использованием табулированных, зависимых от кодека коэффициентов снижения качества 
оборудования для различных коэффициентов потерь пакетов (в предыдущих версиях 
Дополнения I/G.113). С целью снижения объема табулированных данных при использовании 
Е-модели были исследованы возможности замены соответствующими формулами 
табулированных значений Ies при потерях пакетов. Избранный подход привел к результатам, 
близким к тем, которые ранее были определены как значения Ie для всех кодеков, описанных 
в Дополнении I/G.113 версии 2001 года. 

− Предсказания характеристик кодека при зависимой потере пакетов 
 Эта версия алгоритма, учитывающая распределение потерь пакетов, которое характеризуется 

средними (кратковременными) зависимостями (в противоположность долговременным 
зависимостям потерь), была включена в Е-модель. До сих пор такой подход к оценке 
применяли только к кодеку по G.729(A), но, по-видимому, он применим также к кодеку по 
G.723.1, а предположительно и к другим кодекам. В ожидании дальнейших подтверждений 
этот алгоритм не следует использовать с коэффициентами всплесков выше, чем BurstR = 2,0. 
Применять эту модель к коэффициентам всплесков более 2,0 все же можно, если процент 
потерь пакетов Ppl менее 2%. 

– Влияние говорящего на местный эффект  
 Оценки качества речи как функции от STMR для значений > 15 дБ, предложенные в 

предыдущей версии Рекомендации МСЭ-Т G.107 (07/2002), были слишком пессимистичными 
и не точно соответствовали результатам, полученным при слуховых тестах. Это оказалось 
особенно важным для телефонов в Северной Америке, спецификации которой обычно 
предписывают номинальные значения STMR от 16 до 18 дБ. 



 

 Рек. МСЭ-Т G.107 (03/2005)  11 

 В представленной исправленной версии алгоритма Е-модели это замечание отражено 
модификацией соответствующей формулы для Ist как функции от местного эффекта (STMR), 
см. равенство 3-11. 

 Как упоминалось в основном тексте данной Рекомендации, эхо говорящего может сделаться 
более заметным для малошумных значений STMR. Это отражено в равенстве 3-24 заменой 
Idte на Idtes. Чтобы сохранить последовательность, порог эха говорящего был смещен от 
STMR > 15 дБ (G.107, 07/2002) до STMR > 20 дБ (G.107, исправленная версия). Эти 
модификации не повлияли на значения STMR < 15 дБ. Следовательно, предсказание качества 
для коэффициента рейтинга передачи R при установках по умолчанию (STMR = 15 дБ) не 
отличается от предсказания по модели предыдущей версии (07/2002). Величина R по 
умолчанию равна 93,2 для обеих версий – предыдущей и текущей. Такая ситуация изображена 
на рисунке A.2. 

Рисунок A.2/G.107 − Сравнение зависимости R от STMR для текущей  
и предыдущей версий алгоритма Е-модели  

Приложение B 
 

Показатели качества, найденные по коэффициенту рейтинга передачи R 

Диапазон коэффициентов рейтинга передачи R может располагаться от 0 до 100, где R = 0 
представляет крайне низкое качество, а R = 100 представляет очень высокое качество. Е-модель 
обеспечивает статистическую оценку показателей качества. Проценты для решения "хорошо или 
лучше" (GoB) или "плохо или хуже" (PoW) получают из коэффициента R с помощью функции ошибок 
Гаусса: 

  ∫
∞π

=
x t

dtxE
–

2
–

2

e
2
1)( . (B-1) 

Используют следующие равенства: 

  %
16

60–100 





= REGoB . (B-2) 

  %
16
–45100 






= REPoW . (B-3) 

Оценочное значение усредненного мнения (MOSCQE) в диапазоне от 1 до 5 для обычной ситуации 
разговора можно найти по коэффициенту R, используя следующие формулы: 

Для R < 0: 1MOSCQE =  

Для 0 < R < 100: 6
CQE 107)100)(60(035,01MOS −⋅−−++= RRRR  (B-4) 

Для R > 100: 5,4MOSCQE = . 

Эти формулы можно инвертировать в диапазоне 1005,6 ≤≤ R  для вычисления R из MOSCQE  
(см. Дополнение I). Значения GoB, PoW и MOSCQE как функции от R изображены на рисунках B.1 и 
B.2, соответственно. 

Рисунок B.1/G.107 – Значения GoB (хорошо или лучше) и PoW (плохо или хуже)  
как функции от коэффициента рейтинга R 

Рисунок B.2/G.107 – Значения MOSCQE как функции от коэффициента рейтинга R 

В некоторых случаях проектировщики передачи могут быть не знакомы с использованием показателей 
качества, например коэффициента рейтинга R, которые можно получить при проектных расчетах. 
Поэтому для целей проектирования в качестве предварительного руководства для интерпретации 
вычисленных коэффициентов R можно использовать таблицу B.13. В этой таблице также содержатся 
эквивалентные величины R, преобразованные в оценки качества разговора MOSCQE, GoB и PoW. 

Таблица B.1/G.107 – Предварительное руководство для связи величин R  
с удовлетворенностью пользователя 

Величина R 
(нижний 
предел) 

MOSCQE 
(нижний 
предел) 

GoB (%) 
(нижний 
предел) 

PoW (%) 
(верхний 
предел) 

Удовлетворенность  
пользователя 

90 4,34 97 ∼0 Очень удовлетворен 

80 4,03 89 ∼0 Удовлетворен 

70 3,60 73   6 Некоторые пользователи не удовлетворены 
60 3,10 50 17 Многие пользователи не удовлетворены 
50 2,58 27 38 Почти все пользователи не удовлетворены 

____________________ 
3 Источником таблицы B.1 служит таблица 1/G.109 [3]. 

Приложение C 
 

Исходная программа для G.107_5 на языке BASIC 
 
1  CLS 
2  PRINT "PROGRAM g107_4" 
3  REM THIS VERSION IS CONFORM WITH THE ALGORITHM 
4  REM DESCRIBED IN REC. G.107 
5  REM PROGRAM WRITTEN BY N.O. JOHANNESSON 
6  REM MODIFIED BY S. MOELLER, 1999; A. RAAKE, 2003, 2005 
7  PRINT 
8  PRINT "E-model, algorithm according to ITU-T Rec. G.107 (2003) Annex C," 
9  PRINT "for voice communication between side (S) and (R)." 
10 PRINT 
11 PRINT "The E-model gives an estimation of voice transmission quality" 
12 PRINT "mouth-to-ear as perceived at Side (R). The model output is a" 
13 PRINT "transmission rating factor R which can be transformed to give" 
14 PRINT "estimates of customer opinion. Such estimates are only made for" 
15 PRINT "transmission planning purposes and not for actual customer" 
16 PRINT "opinion prediction. See Рек. ITU-T G.107 (2003) for more details." 
17 PRINT 
18 PRINT "Note: To continue program:" 
19 PRINT "a) INPUT appropriate value as indicated on screen," 
20 PRINT "or" 
21 PRINT "b) if no value is asked for, PRESS THE RETURN KEY." 
22 PRINT 
23 DIM X(100), Y(5, 100), Col(5), Pxo(5) 
24 REM Set parameters at default values 
25 GOSUB 29 
26 REM Goto main menu 
27 GOTO 199 
 
 
29 REM Default Parameter Values, according to Tab.3/G.107 (lines 29-60) 
30 SLR = 8 
31 RLR = 2 
32 STMRs = 15 
33 Ds = 3 
34 STMR = 15 
35 Dr = 3 
36 LSTRr = 18 
37 TELR = 65 
38 T = 0 
39 WEPL = 110 
40 Tr = 0 
41 Ta = 0 
42 Ie = 0 
43 Bpl = 1 
44 Ppl = 0 
45 BurstR = 1 
46 A = 0 
47 Nc = -70 
48 Ps = 35 
49 Pr = 35 
50 qdu = 1 
55 Nfor = -64 
60 RETURN 
 
 
198 REM Main menu (lines 198-370) 
199 PRINT 
200 PRINT "Print current parameter values ", "=1" 
210 PRINT "Input new parameter values     ", "=2" 

220 PRINT "Compute table, one parameter   ", "=3" 
230 PRINT "Set parameter at default values", "=4" 
240 PRINT "Exit program                   ", "=5" 
250 PRINT 
260 INPUT Y1 
270 CLS 
280 IF Y1 = 1 THEN GOSUB 500 
290 IF Y1 = 2 THEN GOSUB 1000 
300 IF Y1 = 3 THEN GOSUB 2000 
310 IF Y1 = 4 THEN GOSUB 30 
320 IF Y1 = 5 THEN GOTO 9999 
330 CLS 
340 IF Y1 = 4 THEN PRINT , "Parameters set at default values !" 
350 GOTO 199 
 
500 REM SUB Print current parameter values (lines 500-700) 
510 PRINT , "SLR="; SLR, "RLR="; RLR, "OLR= SLR + RLR="; SLR + RLR 
520 PRINT , "Side (S): Ds="; Ds 
530 PRINT , "Side (R): STMR="; STMR, "Dr="; Dr, "LSTR="; STMR + Dr 
540 PRINT 
550 PRINT , "TELR="; TELR, "Mean One-way Delay T ms="; T 
560 PRINT , "WEPL="; WEPL, "Round-trip Delay Tr ms="; Tr 
570 PRINT , "One-way Absolute Delay Ta ms="; Ta 
580 PRINT 
590 PRINT , "Noise Floor at Side (R) Nfor dBmp="; Nfor 
600 PRINT , "Circuit Noise Nc dBm0p="; Nc 
610 PRINT , "Room Noise, Side (S), Ps dB(A)="; Ps 
620 PRINT , "Room Noise, Side (R), Pr dB(A)="; Pr 
630 PRINT 
640 PRINT , "qdu="; qdu 
650 PRINT 
660 PRINT , "Equipment Impairment Factor Ie="; Ie 
661 PRINT 
662 PRINT , "Packet-loss Robustness Factor Bpl="; Bpl 
663 PRINT 
664 PRINT , "Packet-loss Rate Ppl % ="; Ppl 
665 PRINT 
666 PRINT , "Burst Ratio ="; BurstR 
667 PRINT 
670 PRINT , "Advantage Factor A="; A 
680 PRINT 
690 INPUT C$ 
700 RETURN 
 
1000 REM SUB Input Parameters (lines 1000-1270) 
1020 CLS 
1030 PRINT "Type designation of parameter for which the value is to be changed !" 
1031 PRINT 
1032 PRINT "Note 1. New value of OLR is obtained indirectly, i.e. by new" 
1033 PRINT "value of SLR or RLR. (OLR=SLR+RLR.)" 
1034 PRINT 
1035 PRINT "Note 2. New value of LSTR is obtained indirectly, i.e. by new" 
1036 PRINT "value of STMR or Dr. (LSTR=STMR+Dr.)" 
1037 PRINT 
1040 INPUT "Parameter:"; A$ 
1050 INPUT "New Value="; Px 
1060 PRINT A$; "="; Px 
1070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px 
1080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px 
1090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN STMR = Px 
1100 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN Dr = Px 
1110 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px 
1120 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px 
1130 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px 

1140 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px 
1150 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px 
1160 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px 
1170 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
1171 IF ((A$ = "Bpl") OR (A$ = "BPL") OR (A$ = "bpl")) THEN Bpl = Px 
1172 IF ((A$ = "Ppl") OR (A$ = "PPL") OR (A$ = "ppl")) THEN Ppl = Px 
1173 IF ((A$ = "BurstR") OR (A$ = "BURSTR") OR (A$ = "burstr")) THEN BurstR = Px 
1180 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px 
1190 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px 
1200 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px 
1210 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px 
1220 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px 
1230 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px 
1240 PRINT 
1250 IF Y1 = 2 THEN INPUT "More parameters changed, Yes(1) or No(0)"; Ypar 
1260 IF Ypar = 1 THEN GOTO 1020 
1270 RETURN 
 
2000 REM SUB Tabulate (lines 2000-3000) 
2020 INPUT "Variable Parameter:"; A$ 
2030 PRINT "(To exit tabulation, put parameter value = 1000 !)" 
2040 PRINT TAB(8); A$; TAB(18); "R"; TAB(28); "GOB %"; TAB(38); "POW %"; TAB(48); 
"MOS" 
2050 INPUT Px 
2060 IF Px = 1000 THEN GOTO 3000 
2070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px 
2080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px 
2090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN 
2100   STMR = Px 
2110   LSTR = STMR + Dr 
2120 END IF 
2130 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN 
2140   Dr = Px 
2150   LSTR = STMR + Dr 
2160 END IF 
2170 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px 
2180 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px 
2190 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px 
2200 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px 
2210 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px 
2220 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
2221 IF ((A$ = "Bpl") OR (A$ = "BPL") OR (A$ = "bpl")) THEN Bpl = Px 
2222 IF ((A$ = "Ppl") OR (A$ = "PPL") OR (A$ = "ppl")) THEN Ppl = Px 
2223 IF ((A$ = "BurstR") OR (A$ = "BURSTR") OR (A$ = "burstr")) THEN BurstR = Px 
2230 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px 
2240 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px 
2245 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px 
2250 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px 
2260 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px 
2270 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px 
2280 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
2290 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px 
2300 GOSUB 3500 
2400 GOSUB 4000 
2500 GOSUB 4100 
2600 GOSUB 4200 
2700 R = INT(R * 10 + .5) / 10 
2800 PRINT TAB(8); Px; TAB(18); R; TAB(28); GOB; TAB(38); POW; TAB(48); MOS 
2900 GOTO 2050 
3000 RETURN 
 
3500 REM Compute R (lines 3500-3880) 
 
3509 REM Noise Summation, formulas (3) to (7) 

3510 Nr1 = Ps – SLR – Ds – 100 
3520 Nr1 = Nr1 + .004 * (Ps – SLR – RLR – Ds – 14) ^ 2 
3530 LSTR = STMR + Dr 
3540 Pro = Pr + 10 * LOG(1 + 10 ^ ((10 – LSTR) / 10)) / LOG(10) 
3550 Pr1 = Pro + .008 * (Pro – 35) ^ 2 
3560 Nr2 = Pr1 – 121 + RLR 
3570 Nfo = Nfor + RLR 
3580 No = 10 * LOG(10 ^ (Nr1 / 10) + 10 ^ (Nr2 / 10) + 10 ^ (Nc / 10) + 10 ^ (Nfo 
/ 10)) / LOG(10) 
3590 Nt = No – RLR 
 
3599 REM Ro, formula (2) 
3600 Ro = 15 – 1.5 * (SLR + No) 
 
3609 REM Iolr, formulas (9) and (10) 
3610 Xolr = SLR + RLR + .2 * (64 + Nt) 
3620 Iolr = 20 * ((1 + (Xolr / 8) ^ 8) ^ (1 / 8) – Xolr / 8) 
 
3629 REM Ist, formulas (11) and (12) 
3630 STMRo = -10 * LOG(10 ^ (-STMR / 10) + 10 ^ (-TELR / 10) * EXP(-T / 4)) / 
LOG(10) 
3640 Ist = 12 * (1 + ((STMRo – 13) / 6) ^ 8) ^ (1 / 8) 
3645 Ist = Ist – 28 * (1 + ((STMRo + 1) / 19.4) ^ 35) ^ (1 / 35) 
3650 Ist = Ist – 13 * (1 + ((STMRo – 3) / 33) ^ 13) ^ (1 / 13) + 29 
 
3659 REM Iq, formulas (13) to (17) 
3660 IF qdu < 1 THEN qdu = 1 
3670 Q = 37 – 15 * LOG(qdu) / LOG(10) 
3680 G = 1.07 + .258 * Q + .0602 * Q ^ 2 
3690 Iq = 15 * LOG(1 + 10 ^ ((Ro – 100) / 15) * 10 ^ (46 / 8.4 – G / 9) + 10 ^ 
(46 / 30 – G / 40)) / LOG(10) 
 
3699 REM Is, formula (8) 
3700 Isyn = Iolr + Ist + Iq 
 
3709 REM TERV, formula (22) 
3710 TERV = TELR + 6 * EXP(-.3 * T ^ 2) – 40 * LOG((1 + T / 10) / (1 + T / 150)) 
/ LOG(10) 
3719 REM Modifications to satisfy formula (23) 
3720 IF STMR < 9 THEN TERV = TERV + .5 * Ist 
 
3729 REM Idte, formulas (19) to (21) 
3730 Re = 80 + 2.5 * (TERV – 14) 
3740 Roe = -1.5 * (No – RLR) 
3750 Xdt = (Roe – Re) / 2 
3760 Idte = Xdt + SQR(Xdt ^ 2 + 100) 
3770 Idte = (Idte – 1) * (1 – EXP(-T)) 
 
3779 REM Modifications to satisfy formula (24) 
3780 IF STMR > 20 THEN Idte = SQR(Idte ^ 2 + Ist ^ 2) 
 
3789 REM Idle, formulas (25) and (26)        
3790 Rle = 10.5 * (WEPL + 7) * (Tr + 1) ^ (-1 / 4) 
3800 Xdl = (Ro – Rle) / 2 
3810 Idle = Xdl + SQR(Xdl ^ 2 + 169) 
 
3819 REM Idd, formulas (27) and (28) 
3820 IF Ta < 100 THEN Idd = 0 
3830 IF Ta = 100 THEN Idd = 0 
3840 IF Ta > 100 THEN 
        X = (LOG(Ta / 100)) / LOG(2) 
        Idd = 25 * ((1 + X ^ 6) ^ (1 / 6) – 3 * (1 + (X / 3) ^ 6) ^ (1 / 6) + 2) 
3850 END IF 

3859 REM Id 
3860 Id = Idte + Idle + Idd 
 
3864 REM Inclusion of packet-loss: Ieef, formula (29) 
3865 Ieef = Ie + (95 - Ie) * (Ppl / ((Ppl / BurstR) + Bpl)) 
 
3869 REM R, formula (1) 
3870 R = Ro – Isyn – Id – Ieef + A 
3880 RETURN 
 
4000 REM Compute GOB, formula (B.2) (lines 4000-4050) 
4010 Z# = (R – 60) / 16 
4020 GOSUB 5000 
4030 GOB = 100 * F# 
4040 GOB = INT(GOB * 10 + .5) / 10 
4050 RETURN 
 
4100 REM Compute POW, formula (B.3) (lines 4100-4150) 
4110 Z# = (R – 45) / 16 
4120 GOSUB 5000 
4130 POW = 100 * (1 – F#) 
4140 POW = INT(POW * 10 + .5) / 10 
4150 RETURN 
 
4200 REM Compute MOS, formula (B.4) (lines 4200-4260) 
4210 MOS = 1 + R * .035 + R * (R – 60) * (100 – R) * 7 * 10 ^ (-6) 
4220 MOS = INT(MOS * 100 + .5) / 100 
4230 IF R < 0 THEN MOS = 1 
4240 IF MOS < 1 THEN MOS = 1 
4250 IF R > 100 THEN MOS = 4.5 
4260 RETURN 
 
5000 REM Norm Distr F(Z), formula (B.1) (lines 5000-5130) 
5010 S# = 0 
5020 N% = 0 
5030 H# = Z# 
5040 S# = S# + H# 
5050 H# = H# * (-1) * (Z#) ^ 2 * (2 * N% + 1) / ((N% + 1) * 2 * (2 * N% + 3)) 
5060 N% = N% + 1 
5070 IF ABS(H#) < 10 ^ (-6) THEN GOTO 5090 
5080 GOTO 5040 
5090 S# = S# / (SQR(2 * 3.14159265#)) 
5100 F# = .5 + S# 
5110 F# = INT(F# * 10 ^ 5 + .5) / 10 ^ 5 
5120 REM PRINT "Z="; Z#, "F(Z)="; F#, "N="; N% 
5130 RETURN 
9999 END 

Дополнение I 
 

Вычисление R по значениям MOSCQE 

В диапазоне 1005,6 ≤≤ R  величину R можно вычислить из MOSCQE, используя формулу: 

Функцию arctan2(x, y) используют в ANSI C как функцию atan2(y, x). Пользователи должны заметить, 
что в данном случае порядок этих двух параметров отличается. 

